

Excel für Microsoft 365

Funktionen (Beispiele)



Inhaltsverzeichnis

Einleitung	4
Aufbau einer Excel-Funktion	4
Finanzmathematik	6
EFFEKTIV	6
GDA	6
KAPZ.....	6
LIA	7
RMZ	7
ZINSSATZ	8
ZW	8
Datum und Zeit	9
ARBEITSTAG	9
BRTEILJAHRE.....	9
KALENDERWOCHE.....	10
MONATSENDE	10
NETTOARBEITSTAGE.....	11
STUNDE	11
TAGE	11
WOCHENTAG	12
Mathematisch & Trigonometrisch	12
ABRUNDEN.....	12
ARABISCH.....	12
BASIS.....	13
BOGENMASS	13
FAKULTÄT.....	13
POTENZ.....	13
RÖMISCH.....	14
SUMMEWENN	14
VORZEICHEN	14
ZUFALLSBEREICH	15
Statistik	15
CHIQU.TEST.....	15
F.TEST	15

HÄUFIGKEIT	16
KGRÖSSTE	16
VARIATIONEN	16
ZÄHLENWENN	17
ZÄHLENWENNNS	17
Nachschlage- und Verweisfunktionen (Matrixfunktionen).....	18
FORMELTEXT.....	18
INDEX.....	18
MTRANS	18
SVERWEIS	19
SVERWEIS und VERGLEICH	19
Text.....	21
FINDEN	21
FEST.....	21
GROSS2	21
LÄNGE	22
LINKS	22
SUCHEN	22
TEIL	22
WIEDERHOLEN	23
Logik.....	23
WENN	23
WENNFEHLER.....	24
WENNS.....	24
Informationen.....	24
BLÄTTER	24
FEHLER.TYP	25
INFO.....	25
ISTFORMEL.....	25
ISTGERADE.....	26
TYP	26
ZELLE.....	26
Technik.....	26
DELTA.....	26
HEXINOKT	27

IMWURZEL	27
KOMPLEXE	27
UMWANDELN	28
Hilfe für Funktionen anfordern.....	28

Abbildungsverzeichnis

Abb. 1: <i>Allgemeiner Aufbau einer Funktion in Excel</i>	4
Abb. 2: <i>Beispiel für verschachtelte Funktionen</i>	5
Abb. 3: <i>Funktionen als Argumente für andere Funktionen</i>	5
Abb. 4: <i>Dialogfeld Funktion einfügen</i>	29
Abb. 5: <i>Excel-Hilfe</i>	29
Abb. 6: <i>Quick-Info für Funktionen</i>	29

Einleitung

Der Umgang mit den Excel-Funktionen ist zwar generell nicht besonders kompliziert, allerdings ist nicht immer klar, welche Funktion nun die richtige für die bestehende Aufgabe ist und was die teilweise recht umfangreichen Funktionsargumente bedeuten. In solchen Fällen hilft es meistens nur, die Excel-Hilfe, die Sekundärliteratur oder das Internet zu nutzen, um sich über die einzelnen Funktionen zu informieren. Außerdem ist es nur eine reine Übungssache, sich mit den Funktionen vertraut zu machen. Wenn Sie bestimmte Funktionen öfters einsetzen, werden Sie sie auch ganz schnell beherrschen, so dass Sie sie ganz ohne Funktionsassistent eingeben können. Um den Umgang mit den Excel-Funktionen zu üben, enthält dieses Skript aus verschiedenen Kategorien Beispiele. Die Beispiele sind größtenteils so gehalten, dass Sie keine umfangreichen Daten eingeben müssen, wenn Sie die eine oder andere Funktion selbst ausprobieren wollen. Die hier gezeigten Funktionen sind in [Excel für Microsoft 365](#) verfügbar. Die meistens Beispiele können aber problemlos auch mit älteren Excel-Versionen (auf alle Fälle ab Version 2016) bearbeitet werden.

Aufbau einer Excel-Funktion

Funktionen sind vordefinierte Formeln, die Berechnungen unter Verwendung bestimmter Werte, den sogenannten Argumenten¹, und in einer bestimmten Reihenfolge, auch Syntax genannt, ausführen. Als Argumente können Zahlen, Texte, Wahrheitswerte, Matrizen, Fehlerwerte oder Zellbezüge² verwendet werden. Daneben können Argumente auch aus Konstanten, Formeln oder anderen Funktionen bestehen.

Die Syntax einer Funktion beginnt mit dem Funktionsnamen und nachfolgend einer öffnenden Klammer, den durch Semikola getrennten Argumenten der Funktion und einer schließenden Klammer (siehe Abbildung 1). Allerdings gibt es Funktionen, die kein oder nur ein Argument besitzt. Dann wird auf alle Fälle kein Semikolon benötigt. Die beiden runden Klammern müssen immer angegeben werden, auch wenn die Funktion gar kein Argument besitzt. Steht eine Formel am Anfang einer Funktion, geben Sie ein Gleichheitszeichen (=) vor dem Funktionsnamen ein.




Abb. 1: Allgemeiner Aufbau einer Funktion in Excel

¹ Argumente sind die Werte, die eine Funktion verwendet, um Berechnungen durchzuführen.

² Bei einem Zellbezug handelt es sich um die Koordinaten einer Tabellenzelle oder eines Zellbereichs in einem Arbeitsblatt (z.B. **C4** oder **D3:F12**).

Funktionen können als Argumente für andere Funktionen verwendet werden. Wenn eine Funktion als Argument verwendet wird, d. h. in einer Funktion verschachtelt ist, muss diese den Wertetyp zurückgeben, der von dem Argument verwendet wird. Wenn eine verschachtelte Funktion nicht den richtigen Wertetyp zurückgibt, zeigt Microsoft Excel den Fehlerwert **#WERT!** an.

Beispiel: =FAKULTÄT(ZEICHEN(87))

Die innere Funktion **ZEICHEN** liefert für eine angegebene Zahl (hier: **87**) ein Zeichen des ANSI-Zeichensatz (hier: **87** entspricht dem Zeichen **W**). Die äußere Funktion **FAKULTÄT** benötigt als Argument aber eine Ganzzahl. Das Zeichen **W** ist aber vom Typ *Text*. Daher liefert das Beispiel den Fehlerwert **#WERT!**.

Eine Formel kann bis zu **64** Ebenen verschachtelter Funktionen enthalten. Wenn Funktion *B* als Argument in Funktion *A* verwendet wird, ist Funktion *B* eine Funktion zweiter Ebene. Wenn zusätzlich Funktion *B* als Argument die Funktion *C* enthält, ist Funktion *C* eine Funktion dritter Ebene, usw. Ein Beispiel sehen Sie in Abbildung 2.




Abb. 2: Beispiel für verschachtelte Funktionen

Sie können die Formelpalette verwenden, um Funktionen als Argumente zu verschachteln. Sie können beispielsweise Funktion *B* als ein Argument von Funktion *A* einfügen, indem Sie in der Bearbeitungsleiste auf den Dropdown-Pfeil klicken (siehe Abbildung 3). Wenn Sie für Funktion *A* weitere Argumente eingeben möchten, klicken Sie in der Bearbeitungsleiste auf den Namen von Funktion *A*.




Abb. 3: Funktionen als Argumente für andere Funktionen

Anmerkung: Natürlich muss eine Formel nicht ausschließlich nur aus Excel-Funktionen bestehen. Eine Formel kann auch aus „normalen“ Berechnungen (z.B. Addition oder Multiplikation) und Excel-Funktionen kombiniert werden (z.B. =A7*SUMME(B1:B9)).

Auf den folgenden Seiten finden Sie nun Beispiele aus einigen der Funktionskategorien. Die Beispiele sind teilweise aus der Excel-Hilfe (siehe Kapitel **Hilfe für Funktionen anfordern**, Seite 28) entnommen. Wenn Sie mehr über andere Funktionen wissen wollen, sollten Sie am besten die Excel-Hilfe verwenden. **Noch ein Hinweis:** bei den nachfolgenden Syntaxangaben werden optionale Argumente in eckigen Klammern dargestellt (so wie in den QuickInfos; siehe Abbildung 6, Seite 29).

Finanzmathematik

EFFEKTIV

Gibt die jährliche Effektivverzinsung zurück, ausgehend von einer Nominalverzinsung sowie der jeweiligen Anzahl der Zinszahlungen pro Jahr.

Syntax: **EFFEKTIV(Normalzins;Perioden)**

Beispiel: Angenommen, die Nominalverzinsung beträgt **0,0355** und die Anzahl der Verzinsungsperioden pro Jahr beträgt **4**. Wie hoch ist der Effektivzins?

	A	B	C
1	Nominalzins:	0,0355	
2	Perioden:	4 pro Jahr	
3			
4	Effektivzins:	0,0359754	Formel in B4: =EFFEKTIV(B1;B2)

GDA

Liefert den Abschreibungsbetrag eines Wirtschaftsgutes für eine bestimmte Periode mit Hilfe der Geometrischen Degressiven Abschreibung (GDA) oder einer anderen Abschreibungsmethode.

Syntax: **GDA(Ansch_Wert;Restwert;Nutzungsdauer;Periode;[Faktor])**

Beispiel: Angenommen, eine Firma kauft für **2400 €** eine neue Maschine, die eine Nutzungsdauer von **10 Jahren** haben soll. Der Restwert wird mit **300 €** angesetzt. In den folgenden Beispielen werden für unterschiedliche Perioden die jeweiligen Abschreibungsbeträge berechnet. Die Ergebnisse sind auf zwei Dezimalstellen gerundet.

	A	B	C	D	E	F
1	Wert:	2.400,00 €				
2	Nutzungsdauer:	10 Jahre				
3	Restwert:	300,00 €				
4						
5	1,32 €	Formel in A5: =GDA(B1;B3;B2*365;1)				Die Abschreibung des ersten Tages
6	40,00 €	Formel in A6: =GDA(B1;B3;B2*12;1)				Die Abschreibung des ersten Monats
7	480,00 €	Formel in A7: =GDA(B1;B3;B2;1)				Die Abschreibung des ersten Jahres
8	306,00 €	Formel in A8: =GDA(B1;B3;B2;2;1,5)				Die Abschreibung des zweiten Jahres (Faktor: 1,5)
9	22,12 €	Formel in A9: =GDA(B1;B3;B2;10)				Die Abschreibung des zehnten Jahres

KAPZ

Gibt die Kapitalrückzahlung einer Investition für eine angegebene Periode zurück. Es werden konstante periodische Zahlungen und ein konstanter Zinssatz vorausgesetzt. (**KAPZ** = **KAP**italrück**Zah**lung).

Syntax: **KAPZ(Zins;Zr;Zzr;Bw;[Zw];[F])**

Beispiel: Sie nehmen bei einer Bank einen Kredit in Höhe von **50.000,00 €** auf. Mit der Bank haben Sie eine Laufzeit von **10** Jahren vereinbart. Die Zinsen betragen **7,00%** im Jahr. Sie möchten nun wissen, wie hoch die Rückzahlung im **1. Monat** ist? Außerdem wollen Sie noch wissen, wieviel Sie im **10. Jahr** zurückzahlen müssen?

	A	B	C	D	E	F	G
1	Zinssatz:	7,00%					
2	Zahlungszeitraum:	1					
3	Anzahl der Zahlungszeiträume:	10 Jahre					
4	Kreditsumme (Barwert):	50.000,00 €					
5							
6							
7		-288,88 €	Formel in A7: =KAPZ(B1/12;B2;B3*12;B4)				
8			Der Zinssatz muss durch 12 dividiert werden, da eine monatliche Berechnung durchgeführt werden soll. Auch die Anzahl der Zahlungszeiträume muss mit 12 multipliziert werden, um die Anzahl in Monaten anzugeben.				
9							
10							
11							
12							
13		-6.653,15 €	Formel in A13: =KAPZ(B1;B3;10;B4)				
14			Diesmal wird der Zinssatz und die Anzahl der Zahlungszeiträume nicht durch 12 dividiert oder mit 12 multipliziert, da berechnet werden soll, wie hoch die Rückzahlung für das gesamte 10. Jahr ist.				
15							
16							

LIA

Liefert die lineare Abschreibung eines Wirtschaftsgutes pro Periode (**LIA** = **L**ineare **A**bschreibung).

Syntax: **LIA(Ansch_Wert;Restwert;Nutzungsdauer)**

Beispiel: Angenommen, Sie haben für **35.000 €** einen Lieferwagen gekauft, der eine Nutzungsdauer von **5** Jahren und einen Restwert von **9000 €** hat. Der für jedes Jahr zu berücksichtigende Abschreibungsbetrag ergibt sich wie folgt:

	A	B	C
1	Preis:	35.000,00 €	
2	Nutzungsdauer:	5 Jahre	
3	Restwert:	9.000,00 €	
4			
5	Abschreibungsbetrag:	5.200,00 €	Formel in B5: =LIA(B1;B3;B2)

RMZ

Liefert die konstante Zahlung einer Annuität pro Periode, wobei konstante Zahlungen und ein konstanter Zinssatz vorausgesetzt werden. (**RMZ** = **R**egelmäßige **Z**ahlung).

Syntax: **RMZ(Zins;Zzr;Bw;[Zw];[F])**

Beispiel: Die folgende Anweisung liefert den Betrag, den Sie monatlich zahlen müssen, um bei einem Jahreszinssatz von **8** Prozent einen Kredit von **10.000 €** in **10** Monaten zu tilgen:

	A	B	C
1	jährlicher Zinssatz:	8,00%	
2	Laufzeit:	10 Monate	
3	Kredit:	10.000,00 €	
4			
5	Monatliche Zahlung:	-1.037,03 €	Formel in B5: =RMZ(B1/12;B2;B3)

Sind für denselben Kredit die Zahlungen am Anfang einer Periode fällig, ergibt sich der Betrag entsprechend:

	A	B	C
1	jährlicher Zinssatz:	8,00%	
2	Laufzeit:	10 Monate	
3	Kredit:	10.000,00 €	
4			
5	Monatliche Zahlung:	-1.030,16 €	Formel in B5: =RMZ(B1/12;B2;B3;0;1)

Sie können **RMZ** verwenden, um Zahlungen für andere Annuitäten als Darlehen festzulegen. Wenn Sie beispielsweise in **18 Jahren 50.000 €** ansparen, indem Sie jeden Monat einen gleich großen Betrag zurücklegen, können Sie mit Hilfe von **RMZ** bestimmen, wie viel Sie sparen müssen. Wenn Sie zusätzlich annehmen, dass Sie für Ihr Sparguthaben **1,5%** Zinsen erhalten, können Sie mit **RMZ** ausrechnen, wie viel Sie jeden Monat zurücklegen müssen.

	A	B	C
1	jährlicher Zinssatz:	1,50%	
2	Laufzeit:	18 Jahre	
3	Guthaben (am Anfang):	0,00 €	
4	Guthaben (am Ende):	50.000,00 €	
5			
6	Monatliche Rücklage:	-201,78 €	Formel in B6: =RMZ(B1/12;B2*12;B3;B4)

ZINSSATZ

Liefert den Zinssatz eines voll investierten Wertpapiers.

Syntax: **ZINSSATZ**(Abrechnung;Fälligkeit;Anlage;Rückzahlung;[Basis])

Beispiel: Für eine Obligation gelten die folgenden Modalitäten:

Abrechnungstermin: **14. Mai 2023** Fälligkeitstermin: **22. Juli 2024**
 Kapitalanlage: **1.000.000 €** Rückzahlungsbetrag: **1.017.490 €**
 Basis: **Taggenau/360 (≡ 2)**

Der dem Emissionsdisagio der Obligation entsprechende Zinssatz (im 1900-Datumssystem) berechnet sich wie folgt:

	A	B	C
1	Abrechnungstermin:	14.05.2023	
2	Fälligkeitstermin:	22.07.2024	
3	Kapitalanlage:	1.000.000,00 €	
4	Rückzahlungsbetrag:	1.017.490,00 €	
5	Basis:	2	(entspricht Taggenau/360)
6			
7	Zinssatz:	1,4474%	Formel in B7: =ZINSSATZ(B1;B2;B3;B4;B5)

ZW

Liefert den zukünftigen Wert (Endwert) einer Investition. Die Berechnung basiert auf regelmäßigen, konstanten Zahlungen und einem konstanten Zinssatz. (**ZW** = Zukünftiger **W**ert).

Syntax: **ZW**(Zins;Zzr;Rmz;Bw;[F])

Beispiel: Angenommen, Sie möchten Geld für ein Projekt sparen, das in einem Jahr starten soll. Sie zahlen dafür **1000 €** auf ein Sparkonto ein, für das ein Jahreszinssatz von **6%** vereinbart ist, wobei monatlich aufgezinst wird (monatlicher Zinssatz von **6%/12** oder **0,5%**). Zusätzlich beabsichtigen Sie, während der nächsten **12** Monate jeweils am Monatsanfang **100 €** einzuzahlen. Wie hoch wird Ihr Kontostand am Ende der **12** Monate sein?

	A	B	C
1	monatlicher Zinssatz:	0,50%	
2	Laufzeit:	12 Monate	
3	Guthaben (am Anfang):	1.000,00 €	
4	monatliche Einzahlung:	100,00 €	
5			
6	Endwert:	2.301,40 €	Formel in B6: =ZW(B1;B2;-B4;-B3;1)

Datum und Zeit

ARBEITSTAG

Gibt die Datumsangabe als fortlaufenden Tag im Jahr zurück, vor oder nach einer bestimmten Anzahl von Arbeitstagen. Nicht zu den Arbeitstagen gezählt werden Wochenenden sowie die Tage, die als Ferien (Freie Tage) angegeben sind. **ARBEITSTAG** ermöglicht es, Wochenenden oder Ferien auszuschließen, wenn Fälligkeitstermine für Rechnungen, zu erwartende Lieferzeiten oder die Anzahl bereits verstrichener Arbeitstage berechnet werden sollen.

Syntax: **ARBEITSTAG(Ausgangsdatum;Tage;[Freie_Tage])**

Beispiel:

	A	B	C
1	Ausgangsdatum:	18.03.2024	
2	Anzahl Arbeitstage:	34	
3	Freie Tage:	29.03.2024	Karfreitag
4		01.04.2024	Ostermontag
5		01.05.2024	Maifeiertag
6			
7	Fälligkeitstermin:	08.05.2024	Formel in B7: =ARBEITSTAG(B1;B2;B3;B5)

Anmerkung: Verwenden Sie die Funktion **ARBEITSTAG.INTL**, um die fortlaufende Zahl des Datums vor oder nach einer bestimmten Anzahl von Arbeitstagen mit Parametern zu berechnen, um anzugeben, welche und wie viele Tage als Wochenendtage behandelt werden.

BRTEILJAHRE

Wandelt die Anzahl der ganzen Tage zwischen Ausgangsdatum und Enddatum in Bruchteile von Jahren um. Unter Verwendung der Arbeitsblattfunktion **BRTEILJAHRE** können z. B. Laufzeiten von Forderungen oder Verbindlichkeiten besser miteinander verglichen werden.

Syntax: **BRTEILJAHRE(Ausgangsdatum;Enddatum;[Basis])**

Beispiel:

	A	B	C	D	E
1	Anfangsdatum:	31.03.2020			
2	Enddatum:	27.06.2024			
3	Basis:	3	(entspricht Taggenau/365)		
4					
5	Differenz:	4,2438356	Formel in B5: =BRTEILJAHRE(B1;B2;B3)		
6			Die Differenz zwischen Anfangs- und Enddatum		
7			beträgt 4 Jahre und nicht ganz 3 Monate.		

KALENDERWOCHE

KALENDERWOCHE³ liefert eine Zahl, die angibt, in welche Woche des zugehörigen Jahres das angegebene Datum fällt. Wählen Sie für das Argument **Wochenbeginn** den Wert **21** (System 2), welches dem europäischen Wochenummerierungssystem entspricht.

Syntax: **KALENDERWOCHE(Fortlaufende_Zahl;[Zahl_Typ])**

Beispiel:

	A	B	C	D	E	F	G	H	I	J
1	Datum:	09.06.2024								
2										
3		24	Formel in B3: =KALENDERWOCHE(B1)				Wochenanfang: Sonntag (System 1)			
4		23	Formel in B4: =KALENDERWOCHE(B1;11)				Wochenanfang: Montag (System 1)			
5		24	Formel in B5: =KALENDERWOCHE(B1;14)				Wochenanfang: Donnerstag (System 1)			
6		23	Formel in B6: =KALENDERWOCHE(B1;21)				Wochenanfang: Montag (System 2)			
7										
8		System 1:	Die Woche, die den 1. Januar umfasst, ist die erste Kalenderwoche mit der Nummer 1.							
9										
10		System 2:	Die Woche, die den ersten Donnerstag des Jahres umfasst, ist die Kalenderwoche mit der Nummer 1. Dieses System entspricht der in ISO 8601 spezifizierten Methode, welche im Allgemeinen als europäisches Wochenummerierungssystem bezeichnet wird.							

MONATSENDE

Gibt die fortlaufende Zahl des letzten Tages des Monats zurück, der eine bestimmte Anzahl von Monaten vor bzw. nach dem Ausgangsdatum liegt. Mit **MONATSENDE** können Sie Rückzahlungs- oder Fälligkeitstermine berechnen, die auf den letzten Tag eines Monats fallen. Das Ergebnis der Funktion ist die Anzahl der vergangenen Tage seit dem 01.01.1900. Um das Ergebnis der Funktion als Datum darstellen zu lassen, muss es mit einem passenden Zahlenformat formatiert werden.

Syntax: **MONATSENDE(Ausgangsdatum;Monate)**

Beispiel:

	A	B
1	16.05.2024	
2	31.05.2024	Formel in A2: =MONATSENDE(\$A\$1;0)
3	30.06.2024	Formel in A3: =MONATSENDE(\$A\$1;1)
4	31.07.2024	Formel in A4: =MONATSENDE(\$A\$1;2)
5	30.04.2024	Formel in A5: =MONATSENDE(\$A\$1;-1)
6	31.01.2024	Formel in A6: =MONATSENDE(\$A\$1;-4)

NETTOARBEITSTAGE

Gibt die Anzahl der Arbeitstage in einem Zeitintervall zurück. Nicht zu den Arbeitstagen gezählt werden Wochenenden sowie die Tage, die als Ferien (Feiertage) angegeben sind. Mit **NETTOARBEITSTAGE** können Sie beispielsweise die für Arbeitnehmer zu zahlenden Leistungen berechnen, die auf der zu einem bestimmten Zeitraum gehörenden Anzahl an Arbeitstagen basieren.

Syntax: **NETTOARBEITSTAGE(Ausgangsdatum;Enddatum;[Freie_Tage])**

Beispiel: Es sollen die Nettoarbeitstage zwischen dem **1. März 2024** und dem **31. Juli 2024** berechnet werden. In diesem Zeitraum gibt es einige Feiertage (**29. März 2024, 1. April 2024, 1. Mai 2024, 9. Mai 2024, 20. Mai 2024** und **30. Mai 2024**).

	A	B	C
1	Ausgangsdatum:	01.03.2024	
2	Enddatum:	31.07.2024	
3	Feiertage:	29.03.2024	Karfreitag
4		01.04.2024	Ostermontag
5		01.05.2024	Maifeiertag
6		09.05.2024	Christi Himmelfahrt
7		20.05.2024	Pfingstmontag
8		30.05.2024	Fronleichnam
9			
10	Nettoarbeitstage:	103	Formel in B10: =NETTOARBEITSTAGE(B1;B2;B3:B8)

Anmerkung: Wenn Sie die Anzahl der vollen Arbeitstage zwischen zwei Datumsangaben berechnen möchten, wobei mithilfe von Parametern angegeben wird, welche und wie viele Tage auf Wochenenden fallen, verwenden Sie die Funktion **NETTOARBEITSTAGE.INTL**.

STUNDE

Wandelt eine fortlaufende Zahl in eine Stunde um. Die Stunde wird als ganze Zahl ausgegeben, die einen Wert von 0 (0 Uhr) bis 23 (23 Uhr) annehmen kann.

Syntax: **STUNDE(Zahl)**

Beispiel:

	A	B	C	D	E	F
1	0,378	9	Formel in B1: =STUNDE(A1)			
2	234,599	14	Formel in B2: =STUNDE(A2)			Die Zahl vor dem Dezimaltrennzeichen ist ohne Bedeutung
3	16:23:21	16	Formel in B3: =STUNDE(A3)			

TAGE

Liefert die Anzahl der Tage zwischen zwei Datumsangaben.

Syntax: **TAGE(Zieldatum;Ausgangsdatum)**

Beispiel:

	A	B	C
1	Zieldatum:	02.08.2024	
2	Ausgangsdatum:	01.02.2024	
3			
4	Anzahl der Tage:	183	Formel in B4: =TAGE(B1;B2)

Anmerkung: Anstelle der Funktion **TAGE** liefert eine einfache Subtraktion dasselbe Ergebnis. Im vorliegenden Beispiel kann die Formel in der Zelle **B4** lauten: **=B1-B2**

WOCHENTAG

Wandelt eine fortlaufende Zahl in einen Wochentag um. Der Tag wird standardmäßig als ganze Zahl ausgegeben, die einen Wert von **1** (Sonntag) bis **7** (Samstag) annehmen kann.

Syntax: **WOCHENTAG(Zahl;[Typ])**

Beispiel:

	A	B	C	D
1	16.05.2024			
2				
3	5	Formel in A3: =WOCHENTAG(A1)	1 (Sonntag) bis 7 (Samstag)	5 = Donnerstag
4	4	Formel in A4: =WOCHENTAG(A1;2)	1 (Montag) bis 7 (Sonntag)	4 = Donnerstag
5	3	Formel in A5: =WOCHENTAG(A1;3)	0 (Montag) bis 6 (Sonntag)	3 = Donnerstag
6	2	Formel in A6: =WOCHENTAG(A1;13)	1 (Mittwoch) bis 7 (Dienstag)	2 = Donnerstag
7	6	Formel in A7: =WOCHENTAG(A1;16)	1 (Samstag) bis 7 (Freitag)	6 = Donnerstag

Mathematisch & Trigonometrisch

ABRUNDEN

Eine Zahl mit Dezimalstellen wird auf eine angegebene Anzahl von Dezimalstellen abgerundet.

Syntax: **ABRUNDEN(Zahl;Anzahl_Stellen)**

Beispiel:

	A	B	C
1	Zahlen	Abunden auf 2 Dezimalstellen	
2	43,170308	43,17	Formel in B2: =ABRUNDEN(A2;2)
3	91,697248	91,69	Formel in B3: =ABRUNDEN(A3;2)
4	66,511533	66,51	Formel in B4: =ABRUNDEN(A4;2)
5	61,674954	61,67	Formel in B5: =ABRUNDEN(A5;2)
6	83,401702	83,40	Formel in B6: =ABRUNDEN(A6;2)
7	8,3239582	8,32	Formel in B7: =ABRUNDEN(A7;2)
8	47,048292	47,04	Formel in B8: =ABRUNDEN(A8;2)
9	96,839865	96,83	Formel in B9: =ABRUNDEN(A9;2)

ARABISCH

Konvertiert eine römische Zahl (Groß-/Kleinschreibweise ohne Bedeutung) in eine arabische Zahl.

Syntax: **ARABISCH(Text)**

Beispiel:

	A	B	C	D	E
1	Römische Zahl	Arabische Zahl (Dezimalzahl)			
2	XVII	17	Formel in B2: =ARABISCH(A2)		
3	LIX	59	Formel in B3: =ARABISCH(A3)		
4	CLXXVI	176	Formel in B4: =ARABISCH(A4)		
5	DCCLXII	762	Formel in B5: =ARABISCH(A5)		
6	MDCCCLXXIX	1879	Formel in B6: =ARABISCH(A6)		
7	MMXIII	2013	Formel in B7: =ARABISCH(A7)		

BASIS

Wandelt eine Dezimalzahl in eine Textdarstellung mit der angegebenen Basis um. So können Sie (fast) jede Dezimalzahl in eine Zahl mit anderer Basis umwandeln (nicht nur in binär, hexadezimal oder oktal, sondern auch z.B. in Zahlen zur Basis 3 oder 7 oder 12). Das Ergebnis ist auf alle Fälle aus Sicht von Excel Text.

Syntax: **BASIS(Zahl;Basis;[Mindestlänge])**

Beispiel:

	A	B	C	D	E	F	G
1	Dezimalzahl	Basis	Ergebnis				
2	12	2	1100	Formel in C2: =BASIS(A2;B2)			
3	43	3	1121	Formel in C3: =BASIS(A4;B3)			
4	432	16	1B0	Formel in C4: =BASIS(A4;B4)			
5	342424	8	000000001234630	Formel in C5: =BASIS(A5;B5;15)			Mindestlänge: 15 Zeichen
6	16753	2	100000101110001	Formel in C6: =BASIS(A6;B6;15)			Mindestlänge: 15 Zeichen
7	915342	5	000000213242332	Formel in C7: =BASIS(A7;B7;15)			Mindestlänge: 15 Zeichen

BOGENMASS

Wandelt Grad in Bogenmaß (Radiant) um.

Syntax: **BOGENMASS(Winkel)**

Beispiel:

	A	B	C	D
1	Winkel (in Grad)	Winkel (in Bogenmaß)		
2	90	1,5707963	Formel in B2: =BOGENMASS(A2)	entspricht $\pi/2$
3	180	3,1415927	Formel in B3: =BOGENMASS(A3)	entspricht π
4	270	4,7123890	Formel in B4: =BOGENMASS(A4)	entspricht $3\pi/2$
5	45	0,7853982	Formel in B5: =BOGENMASS(A5)	entspricht $\pi/4$

FAKULTÄT

Liefert die Fakultät einer Zahl ($n! = 1*2*3*4*...*n$; $n \geq 0$; $0! = 1$; $1! = 1$).

Syntax: **FAKULTÄT(Zahl)**

Beispiel:

	A	B	C	D	E
1	1	Formel in A1: =FAKULTÄT(0)			entspricht: 0! = 1 (per Definition)
2	1	Formel in A2: =FAKULTÄT(1)			entspricht: 1! = 1 (per Definition)
3	2	Formel in A3: =FAKULTÄT(2)			entspricht: 2! = 1*2
4	6	Formel in A4: =FAKULTÄT(3)			entspricht: 3! = 1*2*3
5	120	Formel in A5: =FAKULTÄT(5)			entspricht: 5! = 1*2*3*4*5
6	3628800	Formel in A6: =FAKULTÄT(10)			entspricht: 10! = 1*2*3*4*5*6*7*8*9*10
7	2,4329E+18	Formel in A7: =FAKULTÄT(20)			entspricht: 20! = 1*2*3*4*... *17*18*19*20
8	9,3326E+157	Formel in A8: =FAKULTÄT(100)			entspricht: 100! = 1*2*3*4*... *97*98*99*100
9	#ZAH!l	Formel in A7: =FAKULTÄT(1000)			entspricht: 1000! = 1*2*3*4*... *997*998*999*1000; Ergebnis zu groß
10	#ZAH!l	Formel in A9: =FAKULTÄT(-100)			entspricht: -100! = kein Ergebnis, da negative Werte nicht erlaubt sind

POTENZ

Gibt als Ergebnis eine potenzierte Zahl zurück, also $Zahl^{Potenz}$ (z.B. $5^3 = 5*5*5 = 125$).

Syntax: **POTENZ(Zahl;Potenz)**

Beispiel:

	A	B	C	D	E
1	Zahl	Exponent	Ergebnis		
2	3	7	2187	Formel in C2: =POTENZ(A2;B2)	entspricht: 3 ⁷
3	2,87	4,76	151,18818	Formel in C3: =POTENZ(A3;B3)	entspricht: 2,87 ^{4,76}
4	7	1,25	11,386036	Formel in C4: =POTENZ(A4;B4)	entspricht: 7 ^{1,25}

RÖMISCH

Wandelt eine arabische Zahl in eine römische Zahl als Text um. Ist die Zahl größer als 3999, zeigt Excel den Fehlerwert **#WERT!** an.

Syntax: RÖMISCH(Zahl;[Typ])

Beispiel:

	A	B	C	D
1	Dezimalzahl	Römische Zahl		
2	7	VII	Formel in B2: =RÖMISCH(A2)	
3	16	XVI	Formel in B3: =RÖMISCH(A3)	
4	94	XCIV	Formel in B4: =RÖMISCH(A4)	
5	182	CLXXXII	Formel in B5: =RÖMISCH(A5)	
6	499	CDXCIX	Formel in B6: =RÖMISCH(A6)	Standarddarstellung der Zahl 499 als römische Zahl
7	499	ID	Formel in B7: =RÖMISCH(A7;4)	Kurzform der Zahl 499 als römische Zahl
8	2020	MMXX	Formel in B8: =RÖMISCH(A8)	
9	19223	#WERT!	Formel in B9: =RÖMISCH(A9)	Zahl zu groß für Darstellung als römische Zahl

SUMMEWENN

Es wird die Summe eines Zellbereichs berechnet, wobei nur die Zahlen in der Summe verwendet werden, die eine bestimmte Bedingung erfüllen. Die Bedingung muss in doppelte Anführungszeichen eingeschlossen werden.

Syntax: SUMMEWENN(Bereich;Suchkriterien;[Summe_Bereich])

Beispiel: Es sollen im Zellbereich **A1:Z1** alle Zahlen, die kleiner als 5 sind, summiert werden (im Beispiel sind die Zahlen, die die Bedingung erfüllen, farbig hervorgehoben).

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z
1	5	6	9	9	9	7	6	2	7	9	2	2	1	9	2	1	2	2	7	6	7	5	6	1	4	9
2																										
3	19																									

VORZEICHEN


Gibt das Vorzeichen einer Zahl zurück. Ist die Zahl größer als 0, liefert die Funktion **VORZEICHEN** das Ergebnis 1. Ist die Zahl gleich 0, wird der Wert 0 zurückgeliefert. Bei einer negativen Zahl liefert die Funktion den Wert -1.

Syntax: VORZEICHEN(Zahl)

Beispiel:

	A	B	C
1	Zahlen	Vorzeichen	
2	0	0	Formel in B2: =VORZEICHEN(A2)
3	8	1	Formel in B3: =VORZEICHEN(A3)
4	-9	-1	Formel in B4: =VORZEICHEN(A4)
5	3	1	Formel in B5: =VORZEICHEN(A5)
6	-6	-1	Formel in B6: =VORZEICHEN(A6)

ZUFALLSBEREICH

Gibt eine ganze Zufallszahl aus dem festgelegten Bereich zurück. Bei jeder Neuberechnung (z.B. mit der Funktionstaste ) des jeweiligen Arbeitsblatts wird eine neue ganze Zufallszahl ausgegeben.

Syntax: **ZUFALLSBEREICH(Untere_Zahl;Obere_Zahl)**

Beispiel:

	A	B
1	249	Formel in A1: =ZUFALLSBEREICH(100;1000)
2	1	Formel in A2: =ZUFALLSBEREICH(-1;1)
3	-6	Formel in A3: =ZUFALLSBEREICH(-10;10)

Statistik

CHIQU.TEST

Liefert die Teststatistik eines Unabhängigkeitstests. **CHIQU.TEST** gibt den Wert der chiquadrierten (χ^2)-Verteilung für die Teststatistik mit den entsprechenden Freiheitsgraden zurück. Mithilfe von χ^2 -Tests können Sie feststellen, ob in Experimenten die Ergebnisse bestätigt werden, die aufgrund von Hypothesen erwartet wurden.

Syntax: **CHIQU.TEST(Beob_Messwerte;Erwart_Werte)**

Beispiel:

	A	B	C	D
1	Männer (Tatsächlich)	Frauen (Tatsächlich)		
2	58	35	Dafür	
3	11	25	Enthaltung	
4	10	23	Dagegen	
5				
6	Männer (Erwartet)	Frauen (Erwartet)		
7	45	43	Dafür	
8	17	18	Enthaltung	
9	17	22	Dagegen	
10				
11	0,001494741	Formel in A11: =CHIQU.TEST(A2:B4;A7:B9)		

F.TEST

Liefert die Teststatistik eines F-Tests. Ein F-Test berechnet die einseitige Wahrscheinlichkeit, dass sich die Varianzen von Matrix1 und Matrix2 nicht signifikant unterscheiden. Mit dieser Funktion können Sie feststellen, ob zwei Stichproben unterschiedliche Varianzen haben. Sind zum Beispiel die Prüfungsergebnisse öffentlicher und privater Schulen bekannt, können Sie ermitteln, ob bei diesen Schultypen ein unterschiedlich breites Notenspektrum üblich ist. Untersucht wird der Unterschied in der Streuung (oder Vielfalt) der Prüfungsergebnisse.

Syntax: **F.TEST(Matrix1;Matrix2)**

Beispiel:

	A	B	C	D	E	F
1	Matrix 1:	6	7	9	15	21
2	Matrix 2:	20	28	31	38	40
3						
4	0,6483178	Formel in A4: =F.TEST(B1:F1;B2:F2)				

HÄUFIGKEIT

Liefert eine Häufigkeitsverteilung als einspaltige Matrix. Verwenden Sie **HÄUFIGKEIT** z.B., um die Prüfungsergebnisse innerhalb bestimmter Ergebnisbereiche zu zählen. Da **HÄUFIGKEIT** eine Matrix liefert, muss die Formel als Matrixformel eingegeben werden (siehe auch Skript **Excel für Microsoft 365 – Funktionen (Matrix) und dynamische Arrays**).

Syntax: **HÄUFIGKEIT(Daten;Klassen)**

Beispiel: Angenommen, die Ergebnisse einer Prüfung sind in einer Tabelle zusammengestellt. Die erreichten Punktzahlen sind **79, 85, 78, 85, 83, 81, 95, 88** sowie **97** und stehen in dieser Reihenfolge in dem Bereich **A2:A10**. Diese spaltenweise angeordneten Prüfungsergebnisse werden als *Daten* übergeben. Die Intervallgrenzen, gemäß derer die Prüfungsergebnisse gruppiert werden sollen, sind ebenfalls spaltenweise angeordnet und werden als *Klassen* übergeben. In diesem Beispiel ist *Klassen* gleich dem Bereich **B2:B4**, der die Werte **70, 79** und **89** enthält. Wenn Sie **HÄUFIGKEIT** z. B. in die Tabellenzellen **C2:C5** als Matrix eingeben, wird die Häufigkeit der Ergebnisse entsprechend der Einstufungsbereiche **0-70, 71-79, 80-89** und über **90** gezählt. Für dieses Beispiel wird vorausgesetzt, dass alle Prüfungsergebnisse als ganze Zahlen vorliegen. Nachdem Sie neben Ihren Daten vier vertikale Tabellenzellen markiert haben, wird die Formel als Matrixformel eingegeben.

	A	B	C	D	E	F	G	H
1	Punkte	Klassen	Häufigkeit					
2	79	70	0	keine Klausur mit 0 bis 70 Punkten				
3	85	79	2	zwei Prüfungsergebnisse mit 71 bis 79 Punkten				
4	78	89	5	fünf Prüfungsergebnisse mit 80 bis 89 Punkten				
5	85		2	zwei Prüfungsergebnisse über 89 Punkte				
6	83							
7	81		Formel in C2:C5: =HÄUFIGKEIT(A2:A10;B2:B4)					
8	95							
9	88		Hinweis:	Zellbereich C2:C5 markieren, Formel eingeben				
10	97			und mit <Strg>+<Umschalten>+<Eingabetaste>				
11				<Ctrl>+<Shift>+<Return> bestätigen.				

KGRÖSSTE

Gibt den k-größten Wert eines Datenbereichs zurück. Mit dieser Funktion können Sie eine Zahl auf Basis ihrer relativen Größe ermitteln.

Syntax: **KGRÖSSTE(Matrix;k)**

Beispiel:

	A	B	C	D	E	F	G	H
1	33	50	45	86	93	71	20	84
2	90	92	44	29	47	55	97	28
3								
4	97	Formel in A4: =KGRÖSSTE(A1:H2;1)			entspricht: größter Wert der Datenreihe			
5	93	Formel in A5: =KGRÖSSTE(A1:H2;2)			entspricht: zweitgrößter Wert der Datenreihe			
6	86	Formel in A6: =KGRÖSSTE(A1:H2;5)			entspricht: fünftgrößter Wert der Datenreihe			

VARIATIONEN

Liefert die Anzahl der Möglichkeiten, um **k** Elemente aus einer Menge von **n** Elementen ohne Zurücklegen zu ziehen. Eine Variation ist eine Kombination ohne Wiederholung, deren interne Anordnung oder Reihenfolge zu berücksichtigen ist (z.B. Pferdewetten). Variationen unterscheiden sich von Kombinationen, deren interne Anordnung unberücksichtigt bleibt (z.B. Zahlen-Lotto 6 aus 49).

Syntax: **VARIATIONEN(n;k)**

Beispiel: Angenommen, Sie möchten berechnen, mit welcher Chance Sie die Gewinnzahl einer Lotterie wählen. Jede Lotteriezahl besteht aus drei Zahlen, von denen jede einen Wert von 0 (Null) bis 99 annehmen kann.

	A	B	C	D	E	F	G
1	n:	100	k:	3		970200	Formel in F1: =VARIATIONEN(B1;D1)

ZÄHLENWENN

Zählt die nichtleeren Tabellenzellen eines Bereichs, deren Inhalt mit den Suchkriterien übereinstimmen.

Syntax: **ZÄHLENWENN(Bereich;Suchkriterien)**

Angenommen, Sie haben eine Liste mit Ortsnamen (die teilweise mehrmals vorkommen) und möchten nun wissen, wie viele unterschiedliche Ortsnamen die Liste enthält. Diesen Wert können Sie mit der Funktion **ZÄHLENWENN** ermitteln. Die Funktion **ZÄHLENWENN** liefert aber nur dann ein Ergebnis, wenn alle Tabellenzellen im zum analysierenden Zellbereich auch Werte enthalten. Enthält der Zellbereich leere Tabellenzellen, liefert die Funktion **ZÄHLENWENN** einen Fehlerwert.

Beispiel:

	A	B	C	D	E	F	G	H
1	Ort		Anzahl unterschiedlicher Ortsname	8				
2	Hamburg							
3	München		Formel in D1: =SUMME(WENN(\$A\$2:\$A\$17="";0;1/ZÄHLENWENN(\$A\$2:\$A\$17;\$A\$2:\$A\$17)))					
4	Köln							
5	Hamburg		Wichtiger Hinweis: Zur Bestätigung der Formel müssen Sie die Tastenkombination <Strg>+<Umschalten>+<Eingabetaste> verwenden. Es handelt sich also um eine Matrix-Funktion, obwohl nur eine Tabellenzelle ausgewählt wurde und auch nur ein Ergebniswert angezeigt wird. Bei dem Datenbereich handelt es sich aber um einen Zellbereich, also um eine Matrix.					
6	Hamburg							
7	Frankfurt							
8	München							
9	Stuttgart							
10	Berlin							
11	Köln							
12	Berlin							
13	München							
14	Düsseldorf		Eigentlich lautet die Formel =SUMME(1/ZÄHLENWENN(\$A\$2:\$A\$17;\$A\$2:\$A\$17)). Falls aber der Zellbereich \$A\$2:\$A\$17 Leerzellen enthalten sollte, bekommen Sie als Gesamtergebnis einen Fehlerwert angezeigt. Daher wird das Ganze noch mit der Funktion WENN auf Leerzellen hin überprüft.					
15	Hamburg							
16	Bremen							
17	Berlin							

ZÄHLENWENNS

Wendet Kriterien auf Tabellenzellen in mehreren Bereichen an und zählt, wie oft alle Kriterien zutreffen.

Syntax: **ZÄHLENWENNS(Kriterienbereich1;Kriterien1;[Kriterienbereich2;Kriterien2];...)**

Beispiel:

	A	B	C	D	E	F	G	H
1	Datenreihe 1:	12	17	23	42	49	87	
2	Datenreihe 2:	12.07.2024	13.07.2024	14.07.2024	15.07.2024	16.07.2024	17.07.2024	
3								
4	5	Formel in A4: =ZÄHLENWENNS(B1:G1;"<50")						
5	3	Formel in A5: =ZÄHLENWENNS(B1:G1;"<50";B1:G1;">20")						
6	1	Formel in A6: =ZÄHLENWENNS(B1:G1;"<20";B2:G2;"<13.07.2024")						
7	1	Formel in A7: =ZÄHLENWENNS(B1:G1;"<"&D1;B2:G2;"<"&C2)						identisch mit Formel in A6

Nachschlage- und Verweisfunktionen (Matrixfunktionen)

FORMELTEXT

Gibt eine Formel als Zeichenfolge zurück.

Syntax: **FORMELTEXT(Bezug)**

Beispiel:

	A	B	C	D	E
1	0,7853982		=BOGENMASS(45)		Formel in C1: =FORMELTEXT(A1)
2	28.11.2013		=DATUM(2013;11;28)		Formel in C2: =FORMELTEXT(A2)
3	11		=MONAT(A2)		Formel in C3: =FORMELTEXT(A3)

INDEX

Gibt einen Wert oder den Bezug zu einem Wert aus einer Tabelle oder einem Bereich zurück. Die Funktion **INDEX** ist in zwei Versionen vorhanden: in der *Matrixversion* und in der *Bezugsversion*.

Syntax: **INDEX(Matrix;Zeile;[Spalte])** Matrixversion
INDEX(Bezug;Zeile;[Spalte];[Bereich]) Bezugsversion

Beispiel:

	A	B	C	D	E
1	Äpfel	Birnen	Pflaumen		
2	Bananen	Kirschen	Kiwis		
3	Orangen	Zitronen	Pfirsiche		
4	Erdbeeren	Aprikosen	Nektarinen		
5					
6	Bananen		Formel in A6: =INDEX(A1:C4;2;1)		2. Zeile; 1. Spalte
7	Nektarinen		Formel in A7: =INDEX(A1:C4;4;3)		4. Zeile; 3. Spalte
8	Zitronen		Formel in A8: =INDEX(A1:C4;3;2)		3. Zeile; 2. Spalte

MTRANS

Mit der Funktion **MTRANS** wird ein vertikaler Zellbereich als horizontaler Bereich zurückgegeben und umgekehrt. Die Funktion **MTRANS** muss als Arrayformel in einem Bereich eingegeben werden, der die gleiche Anzahl von Zeilen bzw. Spalten wie der Quellbereich hat. Mit der Funktion **MTRANS** kann die vertikale und horizontale Ausrichtung eines Arrays oder Bereichs in ein Arbeitsblatt verschoben werden. Beachten Sie auch die Besonderheiten von Matrixfunktionen (siehe Skript **Excel für Microsoft 365 – Funktionen (Matrix) und dynamische Arrays**).

Syntax: **MTRANS(Matrix)**

Beispiel:

	A	B	C	D	E	F	G	H	I
1		23	53	76	17	28			
2	Matrix A:	82	99	47	75	43			
3		53	29	79	52	72			
4									
5		23	82						
6		53	99						
7	Transponierte Matrix A:	76	47						
8		17	75						
9		28	43						

Formel in B5:C9: =MTRANS(B1:F3)
Für die Eingabe der Formel wird der gesamte Bereich B5:C9 markiert. Dann die Formel eingeben und mit der Tastenkombinationen <Strg>+<Umschalten>+<Eingabetaste> bestätigen.

SVERWEIS

Mit der Funktion **SVERWEIS** können Sie nach der ersten Spalte eines Bereichs von Tabellenzellen suchen und dann einen Wert aus einer beliebigen Tabellenzelle in derselben Zeile des Bereichs zurückgeben.

Syntax: **SVERWEIS(Suchkriterium;Matrix;Spaltenindex;[Bereich_Verweis])**

Beispiel: Angenommen Sie haben eine Materialliste, wobei in der Spalte **A** jedes Teil eine eindeutige Identifikationsnummer besitzt. In Spalte **B** steht die Bezeichnung des jeweiligen Teils und in Spalte **C** die vorhandene Anzahl des jeweiligen Teils im Lager. Mit Hilfe der Funktion **SVERWEIS** können Sie nun anhand der Identifikationsnummer sehr einfach und schnell die Bezeichnung bzw. die Anzahl des jeweiligen Teils ermitteln.

	A	B	C	D	E
1	ID-Nummer	Bezeichnung	Anzahl (Lager)		
2	104	Lüsterklemmen 12-polig	269		
3	137	Expressschelle 11 mm	876		
4	155	Steckdosenleiste 4-fach	82		
5	171	Euro-Zuleitung 2 m	482		
6	189	Treppenlichtzeitschalter	37		
7	209	FI-Schalter 40 A	217		
8	279	Automatischer Zeitschaltuhr	59		
9	314	Universal-Schaltnetzgerät	162		
10					
11	ID-Nummer:	155			
12	Bezeichnung:	Steckdosenleiste 4-fach	Formel in B12: =SVERWEIS(B11;A1:C9;2)		
13	Anzahl:	82	Formel in B13: =SVERWEIS(B11;A1:C9;3)		

Mit der Formel in **B12** wird im Zellbereich **A1:C9** zunächst die in **B11** angegebene Identifikationsnummer gesucht und dann der Zelleninhalt in der 2. Spalte (das ist Spalte **B**) zurückgegeben (das dritte Funktionsargument ist **2**). In Tabellenzelle **B13** steht der Zelleninhalt der 3. Spalte (Spalte **C**; das dritte Funktionsargument ist **3**).

Wird in Tabellenzelle **B11** eine Identifikationsnummer eingetragen, die größer ist als die größte vorhandene Identifikationsnummer, so liefert die Funktion **SVERWEIS** als Ergebnis die Bezeichnung bzw. Anzahl des letzten Tabelleneintrags.

Tragen Sie in die Tabellenzelle **B11** eine Identifikationsnummer ein, die nicht im Tabellenbereich **A2:A9** vorkommt, aber auf alle Fälle kleiner als der größte vorhandene Wert und größer als der kleinste vorhandene Wert ist, so liefert die Funktion **SVERWEIS** immer die zugehörigen Daten des nächstkleineren Werts in Spalte **A** (z.B. **B11** hat den Wert **184**, dann liefert **=SVERWEIS(B11;A1:C9;2)** als Ergebnis **Euro-Zuleitung 2 m** und die Berechnung **=SVERWEIS(B11;A1:C9;3)** liefert als Ergebnis **482**).

Steht in Tabellenzelle **B11** ein Wert, der kleiner ist als der kleinste Wert im Zellbereich **A2:A9**, liefert die Funktion **SVERWEIS** als Ergebnis den Fehlerwert **#NV**.

SVERWEIS und VERGLEICH

An einem zweiten Beispiel soll die Funktion **SVERWEIS** in Kombination mit der Funktion **VERGLEICH** gezeigt werden. Es geht dabei um die Suche nach einem Zelleninhalt in einem Zellbereich. Genauer gesagt um den Inhalt des Schnittpunkts bei einer bestimmten Zeile und Spalte innerhalb des Zellbereichs. Zunächst zum Aufbau der Tabelle (Abbildung siehe nachfolgende Seite). In der Spalte **A**,

genauer gesagt im Zellbereich **A7:A21** stehen verschiedene Längen von Stoffbahnen. In der Zeile **6**, genauer gesagt im Zellbereich **B6:K6** befinden sich die verschiedenen Breiten der Stoffbahnen. Und im Zellbereich **B7:K21** stehen die Preise für die einzelnen Stoffteile. Für dieses Beispiel ist die Tabelle recht klein. In der Praxis aber kann sie deutlich umfangreicher sein. Wenn Sie jetzt den Preis eines Stoffteils für eine bestimmte Länge und Breite ermitteln wollen, könnten Sie natürlich einfach schauen, in welcher Tabellenzelle steht der Wert für die Länge und in welcher Tabellenzelle der Wert für die Breite. Dann müssen Sie nur den Schnittpunkt bestimmen und schon haben Sie den Preis. Bei einer kleinen Tabelle ist das sicherlich kein großes Problem. Aber wie sieht es bei einer sehr großen Tabelle aus? Wenn es sich um hunderte von Zeilen bzw. Spalten handelt. In diesem Fall dürfte es nicht so einfach sein, den gesuchten Preis in der Tabelle auf die Schnelle zu ermitteln. Sie können aber Excel weitestgehend die Arbeit überlassen. Dazu benötigen Sie nur die bereits erwähnten Funktionen **SVERWEIS** und **VERGLEICH**.

	A	B	C	D	E	F	G	H	I	J	K
2	Länge	Breite	Preis								
3	80	40	10,80 €	Formel in C3: =SVERWEIS(A3;A6:K21;VERGLEICH(B3;A6:K6;0);FALSCH)							
5		Breite (in cm)									
6	Länge (in cm)	20	40	60	80	100	120	140	160	180	200
7	20	1,35 €	2,70 €	4,05 €	5,40 €	6,75 €	8,10 €	9,45 €	10,80 €	12,15 €	13,50 €
8	40	2,70 €	5,40 €	8,10 €	10,80 €	13,50 €	16,20 €	18,90 €	21,60 €	24,30 €	27,00 €
9	60	4,05 €	8,10 €	12,15 €	16,20 €	20,25 €	24,30 €	28,35 €	32,40 €	36,45 €	40,50 €
10	80	5,40 €	10,80 €	16,20 €	21,60 €	27,00 €	32,40 €	37,80 €	43,20 €	48,60 €	54,00 €
11	100	6,75 €	13,50 €	20,25 €	27,00 €	33,75 €	40,50 €	47,25 €	54,00 €	60,75 €	67,50 €
12	120	8,10 €	16,20 €	24,30 €	32,40 €	40,50 €	48,60 €	56,70 €	64,80 €	72,90 €	81,00 €
13	140	9,45 €	18,90 €	28,35 €	37,80 €	47,25 €	56,70 €	66,15 €	75,60 €	85,05 €	94,50 €
14	160	10,80 €	21,60 €	32,40 €	43,20 €	54,00 €	64,80 €	75,60 €	86,40 €	97,20 €	108,00 €
15	180	12,15 €	24,30 €	36,45 €	48,60 €	60,75 €	72,90 €	85,05 €	97,20 €	109,35 €	121,50 €
16	200	13,50 €	27,00 €	40,50 €	54,00 €	67,50 €	81,00 €	94,50 €	108,00 €	121,50 €	135,00 €
17	220	14,85 €	29,70 €	44,55 €	59,40 €	74,25 €	89,10 €	103,95 €	118,80 €	133,65 €	148,50 €
18	240	16,20 €	32,40 €	48,60 €	64,80 €	81,00 €	97,20 €	113,40 €	129,60 €	145,80 €	162,00 €
19	260	17,55 €	35,10 €	52,65 €	70,20 €	87,75 €	105,30 €	122,85 €	140,40 €	157,95 €	175,50 €
20	280	18,90 €	37,80 €	56,70 €	75,60 €	94,50 €	113,40 €	132,30 €	151,20 €	170,10 €	189,00 €
21	300	20,25 €	40,50 €	60,75 €	81,00 €	101,25 €	121,50 €	141,75 €	162,00 €	182,25 €	202,50 €

Um jetzt den Preis einer Stoffbahn für eine bestimmte Länge und Breite zu ermitteln, werden die gewünschten Werte in diesem Beispiel in die Tabellenzellen **A3** und **B3** eingetragen. In der Tabellenzelle **C3** geben Sie dann folgende Formel ein:

=SVERWEIS(A3;A6:K21;VERGLEICH(B3;A6:K6;0);FALSCH)

Beim ersten Argument der Funktion **SVERWEIS** geben Sie den Namen der Tabellenzelle an (im Beispiel die Tabellenzelle **A3**), in dem sich der Längenwert befindet, nach dem gesucht werden soll. Als Beispiel wird der Wert **80** (cm) genommen. Mit dem zweiten Argument wird der Zellbereich angegeben, wo der gesuchte Wert sich befindet (in diesem Beispiel der Zellbereich **A6:K21**). Für das dritte Argument benötigt die Funktion **SVERWEIS** die Spaltenbezeichnung. Allerdings sind im Grunde in den Spalten **B** bis **K** nur die Breitenwerte für die Stoffbahnen angegeben. Die können Sie aber nicht als Argument für die Funktion **SVERWEIS** verwenden. Um den Spaltennamen (genauer die Spaltennummer) zu ermitteln, wird die Funktion **VERGLEICH** eingesetzt. Ähnlich wie bei der Funktion **SVERWEIS** geben Sie mit dem ersten Argument der Funktion **VERGLEICH** den Zellnamen an, wo sich der Wert befindet, nach dem gesucht werden soll. In diesem Beispiel steht der gesuchte Wert in der Tabellenzelle **B3**. Beim zweiten Argument geben Sie den Zellbereich an, in dem der Inhalt des ersten Arguments gesucht werden soll. Das ist in diesem Beispiel der Zellbereich **A6:K6**. Beim dritten Argument geben Sie an, ob es sich bei dem gesuchten Wert um einen ungefähren oder um einen exakten

Wert handelt. Soll nach einem exakten Wert gesucht werden, geben Sie den Wert **0** als Argument an. Das Ergebnis der Funktion **VERGLEICH** ist die Nummer der Spalte, in der der gesuchte Wert sich befindet (in diesem Beispiel die Spalte **3**). Damit haben Sie das dritte Argument der Funktion **SVERWEIS**. Beim vierten Argument geben Sie noch an, ob es sich um eine exakte Suche oder eine ungefähre Suche handelt. Bei einer exakten Suche müssen Sie den logischen Wert **FALSCH** als Argument angeben. Auf diese Weise ermitteln Sie den Schnittpunkt aus der Längen- und Breitenangabe.

Text

FINDEN

Sucht eine Zeichenfolge innerhalb einer anderen (Groß-/Kleinschreibung wird beachtet) und liefert als Ergebnis die Nummer der Anfangsposition des Suchtextes, beginnend beim linken Zeichen des Textes. Sie können auch die Funktion **SUCHEN** verwenden, um eine Zeichenfolge innerhalb einer anderen zu suchen, aber im Gegensatz zur Funktion **SUCHEN** berücksichtigt die Funktion **FINDEN** die Schreibweise (Groß-/Kleinbuchstaben) der beteiligten Zeichenfolgen und lässt keine Platzhalterzeichen zu.

Syntax: **FINDEN(Suchtext;Text;[Erstes_Zeichen])**

Beispiel:

	A	B	C	D
1	Miriam Meier	1	Formel in B1: =FINDEN("M";A1)	
2	Miriam Meier	6	Formel in B2: =FINDEN("m";A2)	
3	Miriam Meier	8	Formel in B3: =FINDEN("M";A3;3)	Suche ab der 3. Stelle von links

FEST

Formatiert eine Zahl als Text mit einer festen Anzahl von Nachkommastellen. Dabei wird automatisch auf- oder abgerundet. Sie können auch eine negative Anzahl für die Nachkommastellen angeben. Dann wird die Zahl entsprechend vor dem Dezimalkomma gerundet. Außerdem können Sie noch festlegen, ob Tausendertrennzeichen angezeigt werden oder nicht. Bedenken Sie bitte, dass das Ergebnis aus Sicht von Excel Text ist und keine Zahl.

Syntax: **FEST(Zahl;[Dezimalstellen];[Keine_Punkte])**

Beispiel:

	A	B	C	D	E
3	Zahl als Text:	3.657.854,73	Formel in B3: =FEST(B1)		automatisch zwei Nachkommastellen
4		3.657.854,727	Formel in B4: =FEST(B1;3)		drei Nachkommastellen
5		3.657.850	Formel in B5: =FEST(B1;-1)		gerundet auf eine Stelle vor dem Komma
6		3.658.000	Formel in B6: =FEST(B1;-3)		gerundet auf drei Stellen vor dem Komma
7		3657854,7	Formel in B3: =FEST(B1;1;WAHR)		keine Tausenderpunkte

GROSS2

Wandelt den ersten Buchstaben aller Wörter einer Zeichenfolge in Großbuchstaben um. Wandelt alle anderen Buchstaben in Kleinbuchstaben um.

Syntax: **GROSS2(Text)**

Beispiel:

	A	B	C
1	Das ist ein Beispiel	Das Ist Ein Beispiel	Formel in B1: =GROSS2(A1)
2	anEiHeBETrag: 100 €	Anleihebetrag: 100 €	Formel in B2: =GROSS2(A2)
3	eXceL	Excel	Formel in B3: =GROSS2(A3)

LÄNGE

Gibt die Anzahl der Zeichen einer Zeichenfolge zurück.

Syntax: **LÄNGE(Text)**

Beispiel:

	A	B	C
1	Zeichenfolge:	Kalkulationsprogramm	
2			
3	Länge:	20	Formel in B3: =LÄNGE(B1)

LINKS

Die Funktion **LINKS** gibt auf der Grundlage der Anzahl von Zeichen, die Sie angeben, das erste oder die ersten Zeichen in einer Textzeichenfolge zurück.

Syntax: **LINKS(Text;[Anzahl_Zeichen])**

Beispiel:

	A	B	C
1	Zeichenfolge:	Kalkulationsprogramm	
2			
3	Das 1. Zeichen:	K	Formel in B3: =LINKS(B1)
4	Die ersten 11 Zeichen:	Kalkulation	Formel in B4: =LINKS(B1;11)

SUCHEN

Mit der Funktion **SUCHEN** können Sie nach einer Textzeichenfolge innerhalb einer zweiten Zeichenfolge suchen und die Nummer der Anfangsposition der ersten Zeichenfolge ab dem ersten Zeichen der zweiten Zeichenfolge zurückgeben.

Syntax: **SUCHEN(Suchtext;Text;[Erstes_Zeichen])**

Beispiel:

	A	B	C
1	Zeichenfolge:	Kalkulationsprogramm	
2			
3	Suche a ab der 5. Position:	7	Formel in B3: =SUCHEN("a";B1;5)
4	Suche pro ab der 1. Position:	13	Formel in B4: =SUCHEN("pro";B1)

TEIL

Liefert eine bestimmte Anzahl Zeichen einer Zeichenfolge ab der von Ihnen bestimmten Stelle.

Syntax: **TEIL(Text;Erstes_Zeichen;Anzahl_Zeichen)**

Beispiel:

	A	B	C	D
1	Das ist ein Beispiel für einen Satz in Excel 2019.			
2				
3	Das ist ein	Formel in A3: =TEIL(A1;1;11)		
4	Beispiel	Formel in A4: =TEIL(A1;13;8)		
5	iel für ei	Formel in A5: =TEIL(A1;18;10)		
6	Excel 2019.	Formel in A6: =TEIL(A1;40;11)		

WIEDERHOLEN

Wiederholt einen Text so oft wie angegeben.

Syntax: **WIEDERHOLEN(Text;Multiplikator)**

Beispiel:

	A	B	C
1	Text	Wiederholen 5-mal	
2	abc	abcabcabcabcabc	Formel in Zelle B2: =WIEDERHOLEN(A2;5)
3	*	*****	Formel in Zelle B3: =WIEDERHOLEN(A3;5)
4	Text	TextTextTextTextText	Formel in Zelle B4: =WIEDERHOLEN(A4;5)

Logik

WENN

Die Funktion **WENN** gibt einen bestimmten Wert zurück, wenn eine angegebene Bedingung als **WAHR** bewertet wird, und einen anderen Wert, wenn die Bedingung als **FALSCH** bewertet wird. Wird eines der beiden optionalen Funktionsargumente weggelassen (oder beide), zeigt Excel als Ergebnis **WAHR** oder **FALSCH** in der Tabellenzelle an.

Syntax: **WENN(Wahrheitstest;[Wert_wenn_wahr];[Wert_wenn_falsch])**

Beispiel: Gegeben sei eine Liste mit Klausurergebnissen mit Punktzahlen im Bereich von **0** bis **100**. Mit Hilfe der Funktion **WENN** soll nun überprüft werden, ob die erreichte Punktzahl größer oder gleich **75** ist. In diesem Fall soll **Bestanden** (als normaler Text) angezeigt werden. Im anderen Fall (die Punktzahl ist also kleiner als **75**) soll noch überprüft werden, ob die erreichte Punktzahl größer oder gleich **50** ist (also zwischen **50** und **74** liegt). Trifft diese Bedingung zu, zeigt Excel in der Tabellenzelle den konstanten Text **Nachklausur** an. Ist auch diese Bedingung falsch (Punktzahl ist kleiner als **50**), erscheint in der Tabellenzelle der konstante Text **Nicht bestanden**.

	A	B	C
1	79	Bestanden	Formel in B1: =WENN(A1>=75;"Bestanden";WENN(A1>=50;"Nachklausur";"Nicht bestanden"))
2	34	Nichtbestanden	Formel in B2: =WENN(A2>=75;"Bestanden";WENN(A2>=50;"Nachklausur";"Nicht bestanden"))
3	69	Nachklausur	Formel in B3: =WENN(A3>=75;"Bestanden";WENN(A3>=50;"Nachklausur";"Nicht bestanden"))
4	85	Bestanden	Formel in B4: =WENN(A4>=75;"Bestanden";WENN(A4>=50;"Nachklausur";"Nicht bestanden"))
5	12	Nichtbestanden	Formel in B5: =WENN(A5>=75;"Bestanden";WENN(A5>=50;"Nachklausur";"Nicht bestanden"))
6	97	Bestanden	Formel in B6: =WENN(A6>=75;"Bestanden";WENN(A6>=50;"Nachklausur";"Nicht bestanden"))
7	82	Bestanden	Formel in B7: =WENN(A7>=75;"Bestanden";WENN(A7>=50;"Nachklausur";"Nicht bestanden"))
8	74	Nachklausur	Formel in B8: =WENN(A8>=75;"Bestanden";WENN(A8>=50;"Nachklausur";"Nicht bestanden"))
9	77	Bestanden	Formel in B9: =WENN(A9>=75;"Bestanden";WENN(A9>=50;"Nachklausur";"Nicht bestanden"))

In einem weiteren Beispiel stehen in einer Spalte Nettobeträge, von denen jeweils der Mehrwertsteuerbetrag ermittelt werden soll. Dazu wäre eine einfache Multiplikation ausreichend. Es kann aber sein, dass in der einen oder anderen Tabellenzelle durch fehlerhafte Eingabe versehentlich ein Textzeichen mit eingetippt worden ist. In diesem Fall interpretiert Excel den gesamten Inhalt als Text und bei der Multiplikation kommt nicht das gewünschte heraus, sondern eine Fehlerwert (**#WERT!**). Diese Fälle sollen mit der Funktion **WENN** abgefangen werden. Dazu wird für die Bedingung zusätzlich die Funktion **ISTTEXT** verwendet. Diese Funktion liefert als Ergebnis den Wert **WAHR**, wenn der Zellinhalt Text ist, im anderen Fall **FALSCH**.

	A	B	C
1	Nettopreis	MwSt	
2	265,87 €	50,52 €	Formel in B2: =WENN(ISTTEXT(A2);"Keine Zahl";A2*19%)
3	754,88 €	143,43 €	Formel in B3: =WENN(ISTTEXT(A3);"Keine Zahl";A3*19%)
4	23,76 €	4,51 €	Formel in B4: =WENN(ISTTEXT(A4);"Keine Zahl";A4*19%)
5	#12,87 €	Keine Zahl	Formel in B5: =WENN(ISTTEXT(A5);"Keine Zahl";A5*19%)
6	179,88 €	34,18 €	Formel in B6: =WENN(ISTTEXT(A6);"Keine Zahl";A6*19%)
7	432,87 €	82,25 €	Formel in B7: =WENN(ISTTEXT(A7);"Keine Zahl";A7*19%)
8	m125,32 €	Keine Zahl	Formel in B8: =WENN(ISTTEXT(A8);"Keine Zahl";A8*19%)
9	34,99 €	6,65 €	Formel in B9: =WENN(ISTTEXT(A9);"Keine Zahl";A9*19%)

WENNFehler

Gibt einen von Ihnen festgelegten Wert zurück, wenn eine Formel einen Fehler ergibt. Andernfalls wird das Ergebnis der Formel ausgegeben. Verwenden Sie die Funktion **WENNFehler**, um Fehler in einer Formel aufzuspüren und zu beseitigen.

Syntax: **WENNFehler(Wert;Wert_falls_Fehler)**

Beispiel:

	A	B	C	D
1	x	y	$f_x=x/y$	
2	87	0	Division durch 0	Formel in C2: =WENNFehler(A2/B2;"Division durch 0")
3	33	2	16,5	Formel in C3: =WENNFehler(A3/B3;"Division durch 0")
4	22	1	22	Formel in C4: =WENNFehler(A4/B4;"Division durch 0")
5	50	0	Division durch 0	Formel in C5: =WENNFehler(A5/B5;"Division durch 0")
6	12	5	2,4	Formel in C6: =WENNFehler(A6/B6;"Division durch 0")
7	93	7	13,28571429	Formel in C7: =WENNFehler(A7/B7;"Division durch 0")
8	70	9	7,777777778	Formel in C8: =WENNFehler(A8/B8;"Division durch 0")
9	69	3	23	Formel in C9: =WENNFehler(A9/B9;"Division durch 0")

WENNS

Mit einer einzigen Funktion können mehrere Bedingungen abgeprüft werden und entsprechend mehrere Ergebnisse erzeugt werden. Dies ist einfacher, als wenn Sie mehrere **WENN**-Funktionen ineinander verschachteln.

Syntax: **WENNS(Wahrheitstest1;Wert_wenn_wahr;[Wahrheitstest2;Wert_wenn_wahr;...])**

Beispiel:

	A	B	C
1	Wert		
2	68	Wert kleiner als 100	Formel in B2: =WENNS(A2>100;"Wert über 100";A2<=100;"Wert kleiner als 100")
3	24	Wert kleiner als 100	Formel in B3: =WENNS(A3>100;"Wert über 100";A3<=100;"Wert kleiner als 100")
4	137	Wert über 100	Formel in B4: =WENNS(A4>100;"Wert über 100";A4<=100;"Wert kleiner als 100")
5	35	Wert kleiner als 100	Formel in B5: =WENNS(A5>100;"Wert über 100";A5<=100;"Wert kleiner als 100")
6	117	Wert über 100	Formel in B6: =WENNS(A6>100;"Wert über 100";A6<=100;"Wert kleiner als 100")
7	68	Wert kleiner als 100	Formel in B7: =WENNS(A7>100;"Wert über 100";A7<=100;"Wert kleiner als 100")
8	28	Wert kleiner als 100	Formel in B8: =WENNS(A8>100;"Wert über 100";A8<=100;"Wert kleiner als 100")
9	108	Wert über 100	Formel in B9: =WENNS(A9>100;"Wert über 100";A9<=100;"Wert kleiner als 100")

Informationen

BLÄTTER

Gibt die Anzahl der Blätter (Tabellen, Diagramme, usw.) in einem Bezug zurück. Wird kein Funktionsargument angegeben, liefert die Funktion die Gesamtzahl der Blätter in Arbeitsmappe.

Syntax: **BLÄTTER**([Bezug])

Beispiel:

	A	B	C
1	Anzahl der Tabellenblätter in der aktuellen Arbeitsmappe:	65	Formel in B2: =BLÄTTER()

FEHLER.TYP

Gibt eine Zahl zurück, die einem der Fehlerwerte in Microsoft Excel entspricht, oder den Fehlerwert **#NV**, wenn kein Fehler vorhanden ist. Sie können **FEHLER.TYP** in der Funktion **WENN** verwenden, um einen Fehlerwert zu ermitteln und eine Zeichenfolge, beispielsweise eine Meldung, anstelle des Fehlerwerts zurückzugeben.

Syntax: **FEHLER.TYP**(Fehlerwert)

Beispiel:

	A	B	C	D	E	F	G	H
1	#NAME?	Fehler	Formel in B1: =WENN(ISTZAHL(FEHLER.TYP(A1));"Fehler";"Alles OK")					
2	#DIV/0!	Fehler	Formel in B2: =WENN(ISTZAHL(FEHLER.TYP(A2));"Fehler";"Alles OK")					
3	#WERT!	Fehler	Formel in B3: =WENN(ISTZAHL(FEHLER.TYP(A3));"Fehler";"Alles OK")					
4	989	Alles OK	Formel in B4: =WENN(ISTZAHL(FEHLER.TYP(A4));"Fehler";"Alles OK")					
5								
6	Formel in A1: =MAXX(12;34;87)			korrekter Funktionsname MAX (und nicht MAXX)				
7	Formel in A2: =8/0			Division durch 0 (nicht erlaubt in der Mathematik)				
8	Formel in A3: =SUMME("A";45)			Texte (hier: A) ist als Argument bei SUMME nicht erlaubt				
9	Formel in A4: =23*43			korrekte Multiplikation				

INFO

Liefert Informationen zu der aktuellen Betriebssystemumgebung.

Syntax: **INFO**(Typ)

Beispiel:

	A	B
1	145	Formel in A1: =INFO("DATEIENZAHL")
2	D:\Test\	Formel in A2: =INFO("VERZEICHNIS")
3	Windows (32-bit) NT 10.00	Formel in A3: =INFO("SYSVERSION")
4	16.0	Formel in A4: =INFO("VERSION")
5	pcdos	Formel in A5: =INFO("SYSTEM")

ISTFORMEL

Die Funktion **ISTFORMEL** überprüft, ob sich im angegebenen Zellbezug eine Formel befindet. Enthält der Zellbezug eine Formel, liefert die Funktion den Wert **WAHR**, im anderen Fall den Wert **FALSCH**.

Syntax: **ISTFORMEL**(Bezug)

Beispiel:

	A	B	C	D	E
1	1,0471976		WAHR		Formel in C1: =ISTFORMEL(A1)
2	1,0471976		FALSCH		Formel in C2: =ISTFORMEL(A2)
3					
4	Hinweis: Die Tabellenzelle A1 enthält die Formel =BOGENMASS(60),				
5	während in der Tabellenzelle A2 ein konstanter Wert steht.				

ISTGERADE

Gibt **WAHR** zurück, wenn die Zahl gerade ist, oder **FALSCH**, wenn die Zahl ungerade ist. Besitzt die zu überprüfende Zahl Nachkommastellen, werden diese ignoriert.

Syntax: **ISTGERADE(Zahl)**

Beispiel:

	A	B	C	D	E	F	G	H	I
1	Zahl	3	8	2	7,5	4,3	1	9	2
2	Gerade Zahl?	FALSCH	WAHR	WAHR	FALSCH	WAHR	FALSCH	FALSCH	WAHR
3									
4		Formel in B2: =ISTGERADE(B1)							
5		Formel in C2: =ISTGERADE(C1)							
6		Formel in D2: =ISTGERADE(D1)							
7		...							

TYP

Gibt eine Zahl zurück, mit dem Sie den Datentyp eines angegebenen Wertes bzw. des Zellinhalts einer angegebenen Tabellenzelle ermitteln können. Die Funktion **TYP** liefert folgende Ergebnisse: **1** (Zahl), **2** (Text), **4** (logischer Wert), **16** (Fehlerwert), **64** (Matrix).

Syntax: **TYP(Wert)**

Beispiel:

	A	B	C	D
1	4535		1	Formel in B1: =TYP(A1)
2	Beispiel		2	Formel in B2: =TYP(A2)
3	FALSCH		4	Formel in B3: =TYP(A3)
4	#NV		16	Formel in B4: =TYP(A4)

ZELLE

Mit der Funktion **ZELLE** werden Informationen zur Formatierung, zur Position oder zum Inhalt einer Tabellenzelle zurückgegeben.

Syntax: **ZELLE(Infotyp;[Bezug])**

Beispiel:

	A	B	C	D
1	123	\$A\$1	Formel in B1: =ZELLE("Adresse";A1)	
2		8	Formel in B2: =ZELLE("Breite";A1)	
3		w	Formel in B3: =ZELLE("Typ";A1)	Hinweis: w steht für Wert, also Zahl
4	Beispiel	l	Formel in B4: =ZELLE("Typ";A4)	Hinweis: l steht für Label, also Text

Technik

DELTA

Überprüft, ob zwei Werte gleich sind. Die Funktion liefert **1**, wenn **Zahl1 = Zahl2** gilt; andernfalls liefert sie **0**. Mit dieser Funktion können Sie eine Gruppe von Werten filtern. Beispielsweise können Sie durch Aufsummieren mehrerer **DELTA**-Funktionen berechnen, wie viele gleiche Zahlenpaare vorliegen. Diese Funktion wird auch als Kroneckersymbol bezeichnet.

Syntax: DELTA(Zahl1;[Zahl2])

Beispiel: In zwei Spalten sind verschiedenen Werte eingetragen. Mit der Funktion DELTA soll nun ermittelt werden, ob die zwei Zahlen, die in derselben Zeile nebeneinanderstehen, identisch sind.

	A	B	C	D
1	Wert 1	Wert 2	Wert1=Wert2?	
2	4	4	1	Formel in C2: =DELTA(A2;B2)
3	6	2	0	Formel in C3: =DELTA(A3;B3)
4	2	7	0	Formel in C4: =DELTA(A4;B4)
5	7	7	1	Formel in C5: =DELTA(A5;B5)
6	8	1	0	Formel in C6: =DELTA(A6;B6)

HEXINOKT

Wandelt eine hexadezimale Zahl in eine oktale Zahl um.

Syntax: HEXINOKT(Zahl;[Stellen])

Beispiel:

	A	B	C	D
1	F	17	Formel in B1: =HEXINOKT(A1)	
2	3B4E	35516	Formel in B2: =HEXINOKT(A2)	
3	FFFABD3	1777725723	Formel in B3: =HEXINOKT(A3)	
4	DFE287FA	#ZAHL!	Formel in B4: =HEXINOKT(A4)	Hexadezimalzahl ist zu groß für die Umwandlung

IMWURZEL

Gibt die Quadratwurzel einer komplexen Zahl zurück, die als Zeichenfolge der Form $x+yi$ oder $x+yj$ eingegeben wird.

Syntax: IMWURZEL(Komplexe_Zahl)

Beispiel:

	A	B	C
1	7+3i	2,70331029534753+0,554875258893343i	Formel in B1: =IMWURZEL(A1)
2	1+i	1,09868411346781+0,455089860562227i	Formel in B2: =IMWURZEL(A2)
3	2+7i	2,15407867652049+1,62482458888345i	Formel in B3: =IMWURZEL(A3)
4	134+23i	11,6180797210407+0,989836554415541i	Formel in B4: =IMWURZEL(A4)
5	23-5i	4,82375396303956-0,518268555808492i	Formel in B5: =IMWURZEL(A5)

KOMPLEXE

Wandelt den Real- und Imaginärteil in eine komplexe Zahl um ($x+yi$ oder $x+yj$).

Syntax: KOMPLEXE(Realteil;Imaginärteil;[Suffix])

Beispiel:

	A	B	C	D
1	Realteil	Imaginärteil	komplexe Zahl	
2	4	5	4+5i	Formel in C2: =KOMPLEXE(A2;B2)
3	3,534	9	3,534+9i	Formel in C3: =KOMPLEXE(A3;B3)
4	7	4,2524	7+4,2524j	Formel in C4: =KOMPLEXE(A4;B4;"j")
5	23,857	12,67546	23,857+12,67546i	Formel in C5: =KOMPLEXE(A5;B5)

UMWANDELN

Wandelt eine Zahl von einem Maßsystem in ein anderes um.

Syntax: **UMWANDELN(Zahl;Von_Maßeinheit;In_Maßeinheit)**

Beispiel:

	A	B	C	D
1	Von	Nach		
2	17,34	7,865291696	Formel in B1: =UMWANDELN(A1;"lbm";"kg")	von Pfund (USA) in Kilogramm
3	31,7	89,06	Formel in B2: =UMWANDELN(A2;"C";"F")	von Celsius in Fahrenheit
4	132,76	#NV	Formel in B3: =UMWANDELN(A3;"ft";"sec")	von Fuß (USA) in Sekunden

Hilfe für Funktionen anfordern

Wenn Sie weitere Informationen zu den oben angegebenen und allen anderen Funktionen haben wollen, führen Sie folgende Schritte aus:

1. Wählen Sie im Register **Formeln** in der Gruppe **Funktionsbibliothek** den Befehl **Funktion einfügen**. Sie erhalten das Dialogfeld **Funktion einfügen** (siehe Abbildung 4, Seite 29).
2. Wählen Sie im Listenfeld **Kategorie auswählen** die Funktionskategorie aus, wo sich die Funktion befindet, für die Sie Hilfeinformationen haben wollen. Wählen Sie gegebenenfalls die Kategorie **Alle**, wenn Sie sich nicht sicher sind, zu welcher Kategorie die gesuchte Funktion gehört.
3. Wählen Sie in der Liste **Funktion auswählen** die Funktion aus, zu der Sie weitere Informationen benötigen.
4. Klicken Sie auf den Link **Hilfe für diese Funktion**. In dem Fenster **Excel-Hilfe** (siehe Abbildung 5, Seite 29) bekommen Sie dann ausführlichere Informationen zu der Funktion.



Bei dem Hilfe-Fenster handelt es sich um einen ganz normalen Webbrowser (z.B. *Mozilla Firefox*, *Google Chrome*, *Microsoft Edge*), wie er für die Darstellung von Webseiten verwendet wird. Bei den Hilfeseiten für die Excel-Funktionen handelt es sich also um Webseiten, d.h. Sie benötigen eine Internetverbindung, um die Excel-Hilfe überhaupt benutzen zu können. Wenn Sie alle gewünschten Informationen zur Funktion erhalten haben, schließen Sie den Webbrowser. Sofern Sie nur Hilfe für eine Funktion haben wollten, klicken Sie im Dialogfeld **Funktion einfügen** auf die Schaltfläche **Abbrechen**. Wollen Sie die Funktion verwenden, klicken Sie auf die Schaltfläche **OK**. Excel wechselt zum Dialogfeld **Funktionsargumente**, wo Sie noch die Argumente für die Funktion eingeben (können) und bestätigen die Angaben über die Schaltfläche **OK**.

Wenn Sie bereits die Funktion besser kennen und die Formel direkt in die Tabellenzelle eingeben, können Sie trotzdem Hilfe für die Funktion bekommen. Während der Eingabe der Formel wird der Funktionsname zusammen mit den Funktionsargumenten als Quick-Info angezeigt (siehe Abbildung 6, Seite 29). Sie können nun mit der Maus direkt den Funktionsnamen im Quick-Info anklicken und erhalten den Aufgabenbereich **Hilfe**. (allerdings kann es vorkommen, dass im Aufgabenbereich nichts angezeigt wird. In diesem Fall müssen Sie die Hilfe zur ausgewählten Funktion über das Dialogfeld **Funktion einfügen** aufrufen). Der Inhalt des Aufgabenbereichs ist mit dem Inhalt im Webbrowser identisch (siehe Abbildung 5, Seite 29), nur die optische Darstellung ist anders.




Abb. 4: Dialogfeld **Funktion einfügen**





Abb. 5: *Excel-Hilfe*



● **Klicken Sie im Quick-Info auf den Funktionsnamen, um die Excel-Hilfe für die Funktion zu öffnen.**

Abb. 6: *Quick-Info für Funktionen*

Anmerkung: Wird kein Quick-Info für eine Funktion angezeigt, kann das daran liegen, dass die entsprechende Einstellung bei den Excel-Optionen ausgeschaltet ist⁴. Klicken Sie auf das Register **Datei** und wählen den Befehl **Optionen**. Im Dialogfeld **Excel-Optionen** wählen Sie die Kategorie **Erweitert** und aktivieren in der Gruppe **Anzeige** das Kontrollkästchen **QuickInfos für Funktionen anzeigen**.

⁴ Es muss nicht zwangsläufig an der Einstellung in den Excel-Optionen liegen. Für manche Funktionen gibt es generell kein Quick-Info.